

Programação para Computação

Prova 1 – 22/09/2008 – Resolução

Prof. Marcus Vinícius Midena Ramos

Engenharia de Computação – UNIVASF

Questão 1 (1,2 ponto):

Descreva, de forma clara, completa e concisa, a sintaxe e a semântica dos seguintes comandos da linguagem Visualg:

a) enquanto ... fimenquanto

Sintaxe:

enquanto <expr> faça <com> fimenquanto

Semântica:

(i) <expr> é avaliada; (ii) se o seu valor for verdadeiro, <com> é executado e retorna-se ao passo (i); (iii) se o seu valor for falso, o comando enquanto é encerrado.

b) repita ...

Sintaxe:

repita <com> ate <expr>

Semântica:

(i) <com> é executado; (ii) <expr> é avaliada; (iii) se o seu valor for falso, retorna-se ao passo (i); (iv) se o seu valor for verdadeiro, o comando repita é encerrado.

c) para ... fimpara

Sintaxe:

para <var> de <expr1> ate <expr2> [passo <expr3>] faça <com>
fimpara

Semântica:

(i) <var> é inicializada com o valor de <expr1>; (ii) se <var> <= <expr2>, então <com> é executado; <var> <- <var> + <expr3> e retorna-se ao passo (i); (iii) se <var> > <expr2>, então o comando para é encerrado. Nota 1: se <expr3> não for especificada, considera-se o valor 1. Nota 2: se <expr3> resultar num valor negativo, então <var> é decrementada (ao invés de ser incrementada) e <com> é executado apenas se <var> >= <expr3>.

Questão 2 (0,8 ponto):

A conversão de base de um número pode ser feita através de divisões inteiras sucessivas desse número pela base desejada, até que o resultado seja zero. O resto obtido em cada operação corresponde à um novo algarismo do número na nova base: Exemplo de conversão de 125 da base 10 para a base 2:

Número:	Dividido por 2, resulta:	Dividido por 2, resta:
125	62	1
62	31	0
31	15	1
15	7	1
7	3	1
3	1	1
1	0	1

Portanto, $125_{10} = 1111101_2$.

Deseja-se elaborar um algoritmo que aceite como entrada um número inteiro na base 10 e gere na saída o número correspondente na base 2. Preencha as lacunas no algoritmo abaixo para que esse efeito seja alcançado:

```

algoritmo "conversão"
var n, ① _____ : ② _____
inicio
escreval ("Digite um valor inteiro na base 10 (maior ou igual a 0):")
leia (③ _____)
escreval ("O valor correspondente na base 2 é (os algarismos estão na
ordem inversa):")
repita
    resto <- ④ _____
    escreva (⑤ _____:1)
    ⑥ _____ <- ⑦ _____
ate ⑧ _____
escreval
fimalgoritmo

```

Exemplo de entrada e saída correspondente pretendidas para o algoritmo acima:

```

Digite um valor inteiro na base 10 (maior ou igual a 0):
125
O valor correspondente na base 2 é (os algarismos estão na ordem
inversa):
1011111

```

```

algoritmo "conversão"
var n, resto: inteiro
inicio
escreval ("Digite um valor inteiro na base 10 (maior ou igual a 0):")
leia (n)
escreval ("O valor correspondente na base 2 é (os algarismos estão na
ordem inversa):")
repita
    resto <- n%2
    escreva (resto:1)
    n <- n\2
ate n=0
escreval
fimalgoritmo

```

Questão 3 (2 pontos):

Elabore um algoritmo que gere na saída a tabuada dos números de 1 a 9, englobando as quatro operações aritméticas (multiplicação, divisão inteira, soma e subtração), conforme o exemplo abaixo (usar a mesma formatação de tela):

1x1= 1	1/1= 1	1+1= 2	1-1= 0
1x2= 2	1/2= 0	1+2= 3	1-2=-1
1x3= 3	1/3= 0	1+3= 4	1-3=-2
1x4= 4	1/4= 0	1+4= 5	1-4=-3
1x5= 5	1/5= 0	1+5= 6	1-5=-4
1x6= 6	1/6= 0	1+6= 7	1-6=-5
1x7= 7	1/7= 0	1+7= 8	1-7=-6
1x8= 8	1/8= 0	1+8= 9	1-8=-7
1x9= 9	1/9= 0	1+9=10	1-9=-8

...

9x1= 9	9/1= 9	9+1=10	9-1= 8
9x2=18	9/2= 4	9+2=11	9-2= 7
9x3=27	9/3= 3	9+3=12	9-3= 6
9x4=36	9/4= 2	9+4=13	9-4= 5
9x5=45	9/5= 1	9+5=14	9-5= 4
9x6=54	9/6= 1	9+6=15	9-6= 3
9x7=63	9/7= 1	9+7=16	9-7= 2
9x8=72	9/8= 1	9+8=17	9-8= 1
9x9=81	9/9= 1	9+9=18	9-9= 0

```

algoritmo "semnome"
var n,i: inteiro
inicio
para n de 1 ate 9 faca
    para i de 1 ate 9 faca
        escreva (n:1,"x",i:1,"=",n*i:2," ")
        escreva (n:1,"/",i:1,"=",n/i:2," ")
        escreva (n:1,"+",i:1,"=",n+i:2," ")
        escreva (n:1,"-",i:1,"=",n-i:2," ")
        escreval
    fimpara
    escreval
fimpara
fimalgoritmo

```

Questão 4 (2 pontos):

Elabore um algoritmo que aceite como entrada:

1. Três números inteiros maiores do que zero (valores que serão representados no gráfico);
2. Um número inteiro maior do que zero (número de colunas do gráfico).

E gere como saída um gráfico de barras, no qual o maior dos três números informados no item 1) será representado com a quantidade de colunas informada no item 2) acima, e as demais com uma quantidade de colunas proporcional aos respectivos valores. Exemplo de entradas e correspondentes saídas (o mesmo formato de tela deverá ser gerado pelo seu algoritmo):

Digite três números inteiros positivos:

2

8

20

Digite o número de colunas:

10

Gráfico de colunas:

*

```

algoritmo "semnome"
var i,col,n1,n2,n3,maior: inteiro
inicio
escreval ("Digite três números inteiros positivos:")
leia (n1,n2,n3)
escreval ("Digite o número de colunas:")
leia (col)
maior <- n1
se n2 > maior entao
    maior <- n2
fimse
se n3 > maior entao

```

```

    maior <- n3
fimse
escreval ("Gráfico de colunas:")
para i de 1 ate n1*col\maior faca
    escreva ("*")
fimpara
escreval
para i de 1 ate n2*col\maior faca
    escreva ("*")
fimpara
escreval
para i de 1 ate n3*col\maior faca
    escreva ("*")
fimpara
escreval
fimalgoritmo

```

Questão 5 (2 pontos):

Elabore um algoritmo que aceite como entrada três números inteiros representando, nesta ordem, um ano, um mês e um dia. O algoritmo deverá validar a entrada, conforme os seguintes critérios:

1. O ano deverá ser um valor entre 1900 (inclusive) e 2100 (inclusive);
2. O mês deverá ser um valor entre 1 e 12;
3. O dia deverá ser um valor entre 1 o número de dias do mês informado (desconsiderar anos bissextos).

Caso um ou mais dos valores digitados sejam inválidos, o algoritmo deverá efetuar nova entrada de dados, procedendo assim até que todos os valores digitados pelo usuário sejam válidos segundo os critérios acima. Quando isso acontecer, o algoritmo deverá emitir uma mensagem validando a data informada. A tela de entrada e saída de dados deverá seguir o formato:

```

Digite o ano (entre 1900 e 2100):
2008
Digite o mês (entre 1 e 12):
9
Digite o dia (conforme o número de dias do mês):
22
A data 22/9/2008 é válida.

```

```

algoritmo "semnome"
var dia, mes, ano, totaldias: inteiro
inicio
repita
escreval ("Digite o ano (entre 1900 e 2100):")
leia (ano)
ate (ano>=1900) e (ano<=2100)
repita
escreval ("Digite o mês (entre 1 e 12):")
leia (mes)
ate (mes>=1) e (mes<=12)
escolha mes
    caso 2
        totaldias <- 28
    caso 1,3,5,7,8,10,12
        totaldias <- 31
    outrocaso
        totaldias <- 30
fimescolha
repita

```

```

escreval ("Digite o dia (conforme o número de dias do mês):")
leia (dia)
ate (dia>=1) e (dia<=totaldias)
escreval ("A data ",dia:1,"/",mes:1,"/",ano:4," é válida.")
fimalgoritmo

```

Questão 6 (2 pontos):

Elabore um algoritmo que aceite como entrada um número inteiro positivo $n \geq 1$ e gere, como saída, o valor real correspondente à somatória dos termos abaixo (o valor de n não precisa, necessariamente, ser múltiplo de 3):

$$\frac{1}{1^1} + \frac{1}{2^2} + \frac{1}{3^3} + \frac{1}{4^1} + \frac{1}{5^2} + \frac{1}{6^3} + \frac{1}{7^1} + \dots + \frac{1}{(n-2)^1} + \frac{1}{(n-1)^2} + \frac{1}{n^3}$$

Exemplo de entrada e saída:

```

Digite um valor inteiro (maior ou igual a 1):
5
O resultado é: 1.57703703703704

```

```

algoritmo "semnome"
var n, i: inteiro
    soma: real
inicio
escreval ("Digite um valor inteiro (maior ou igual a 1):")
leia (n)
soma <- 1
para i de 2 ate n faca
    se i%3=0 entao
        soma <- soma + 1/(i*i*i)
    fimse
    se i%3=1 entao
        soma <- soma + 1/i
    fimse
    se i%3=2 entao
        soma <- soma + 1/(i*i)
    fimse
fimpara
escreval ("O resultado é: ", soma)
fimalgoritmo

```